




හයවන ලිපිය - දෙවන කොටස (II)

## සරල ගණක පරිපථයක් ගොඩ නැගීම

සථන ඛණ්ඩ ප්‍රදර්ශක (Seven Segment Displays) හඳුරුවන ආකාරය පිළිබඳ මූලික කරුණු කිහිපයක් පසුගිය ලිපියෙන් ඉදිරිපත් කෙරී ඇති අතර එ මත පදනම් ව බොත්තමක් එබූ විට අගය 1කින් වැඩි වන ආකාරයේ ගණක පරිපථයක් ගොඩනගන අයුරු විස්තර කිරීම මේ ලිපියේ අරමුණ වේ.

රූප සටහන අංක 1න් දැක්වෙන පරිපථයේ PIC 16F84 (A) මයික්‍රොකොන්ට්‍රොලයරට සම් කළ සථන ඛණ්ඩ දර්ශකය හා සම්චය මෙහි දී විශේෂයෙන් වැදගත් වේ. පරිපථයට විදුලිය සැපයූ මොහොතේ ම සථන ඛණ්ඩ ප්‍රදර්ශකයේ ඕනෑම දිස් විය යුතු අතර බොත්තම (sw1 සම්චය) වරක් එබූ විට අංක එක දිස් විය යුතු අතර නැවත වරක් බොත්තම එබූ විට අංක දෙක "2" දිස් විය යුතු ය. මෙසේ බොත්තම ඔබන වාරයක් පාසා අගය එකකින් වැඩි වී අංක 9ට පැමිණී පසු නැවත "0"න් පටන් ගත යුතු ය.



රූප සටහන අංක 1

රූප සටහන අංක 2 මගින් අදාළ ක්‍රමලේඛනය දැක්වේ. මෙහි දී PORTBහි RB0-RB6 අග්‍ර සථන ඛණ්ඩ ප්‍රදර්ශක ධාවනය කිරීමට යොදා ඇති නිසා එවා ප්‍රතිදන ලෙසත් RA0 අග්‍රය sw1 සම්චයට සම්බන්ධ කර එමගින් සංඥාවක් ලබාගැනීමට යොදා ගන්නා නිසා ප්‍රදන අග්‍රයක් ලෙසත් සකස් කළ යුතු ය. Clrf TRISB හා bsf TRISA, 0 මගින් එම සකස් කිරීම් සිදු කර ඇත. (ප්‍රදන හා ප්‍රතිදන ලෙස අග්‍ර සකස් කිරීම 4 වන ලිපියේ සඳහන් විය)

පටන් ගැනීමේ දී බිංදුව ප්‍රදර්ශනය කිරීම සඳහා ඊට අදාළව PORTB රෙජිස්ටරයට ලිවිය යුතු අගය වන 00111111 මුලින් ම w රෙජිස්ටරයට ලියා ඉන් පසුව එය PORTB රෙජිස්ටරයට මාරු කර ඇත. Morlw b'00111111 හා movwf PORTB යන උපදෙස් දෙක මගින් එම කාර්යයන් සිදු කෙරේ.

ඉන්පසු බොත්තම ඔබන තෙක් බලා සිටිය යුතු ය. බොත්තම එබූ විට RA0 අග්‍රයේ අගය හෙවත් PORTA රෙජිස්ටරයේ බිංදුව වැඩි බිටුව "0" වේ.

නිවැරැදි කිරීමක්

පසුගිය පාඩමේ රූප සටහන අංක 6 ලෙස පළ වී තිබූ රූපයෙහි කොටසක් මුද්‍රණය වී නොතිබිණි. එම රූපය නිවැරැදිව පහත පළ වේ.

btfsc PORTA,0 යන උපදෙසෙන් කියවෙනුයේ PORTA රෙජිස්ටරයේ බිංදුව වැඩි බිටුව "0" දූ ශි බලා එසේ වී නම් ඊළඟට ඇති උපදෙස හෙවත් go to Loop යන්න මඟහැර අගය 1කින් වැඩි කරන incf count, 1 යන උපදෙස ක්‍රියාත්මක කරන ලෙසයි. එම නිසා බොත්තම එබූ විගස මෙම උපදෙසට පැමිණි count නැමැති විචල්‍යයේ අගය එකකින් වැඩි කර ගනී. ඉන්පසු count විචල්‍යයේ අගය දහය වී දූ ශි බැලීම සඳහා movlw D'10 හා xoRwf count, 0 යන උපදෙස් දෙක භාවිත කර ඇත. මෙහි දී කෙරෙනුයේ මුලින් ම w රෙජිස්ටරයට දහයේ ඉලක්කම ගෙන count විචල්‍යයේ අගය එ හා සමාන දූශි බැලීමයි. අගයන් දෙක සමාන නම් STATUS රෙජිස්ටරයේ දෙවන බිටුව "1" බවට පත් වේ. අගයන් සමාන නැතිනම් එම බිටුව "0" වේ. මෙසේ වනුයේ ඉහත XOR නමැති තාර්කික ක්‍රියාවේ දී ප්‍රතිඵලය බිංදුව වීම හෝ නොවීම අනුවයි. Count විචල්‍යයේ අගය දහය වී නම් STATUS රෙජිස්ටරයේ දෙවන බිටුව "1" වන බැවින් btfsc STATUS S 2 යන උපදෙසට පසුව ඇති clrf count යන උපදෙස් මඟහරේ. එවිට count විචල්‍යයේ තිබූ අගය එලෙස ම පවතී.

දැන් එම අගය සථන ඛණ්ඩ ප්‍රදර්ශකය මගින් දැක්විය යුතු නිසා ඊට අදාළ ව PORTB රෙජිස්ටරයට ලිවිය යුතු අගයන් Table ලෙස නම් කර ඇති කොටසෙන් දැක්වේ. මෙවැනි වගුවක් Look up table ලෙස තාක්ෂණික ව්‍යවහාරයේ දී හැඳින්වේ.

මෙම වගුව භාවිත කිරීමට මුලින් ම අදාළ ඉලක්කම w රෙජිස්ටරයට දැමිය යුතු ය. movf conut, 0 මගින් count විචල්‍ය හෙවත් ප්‍රදර්ශනය කළ යුතු අංකය W රෙජිස්ටරයට ලබා දේ. ඉන්පසු call Table යන උපදෙසින් අදාළ වගුවට යන ලෙස දන්වා සිටී. මේ වන විටත් w රෙජිස්ටරයේ අදාළ අංකය එලෙස ම පවතින බැවින් Addwf PC,1 යන උපදෙසෙන් කියවෙන පරිදි එම ඉලක්කම PC රෙජිස්ටරයට එකතු කෙරේ. එකතු කරන ඉලක්කම අනුව අදාළ තීරුව වෙත ගොස් එහි ඇති අගය W රෙජිස්ටරයට ද ගෙන නැවත ආපසු call Table යන උපදෙසට පසුව ඇති movwf portb උපදෙස වෙත පැමිණේ. උදාහරණයක් ලෙස ප්‍රදර්ශනය කළ යුතු අගය "5" නම් ඊට අදාළ retlw b' 01101101" යන උපදෙසට පැමිණ එහි ඇති අගය වන 01101101 නමැති දෙකේ පාදයේ සංඛ්‍යාව w රෙජිස්ටරයට ගෙන ආපසු movwf PORTB උපදෙසට ගමන් කරයි.

එහි දී W රෙජිස්ටරයේ තිබූ අගය PORTB රෙජිස්ටරයට මාරු කරන අතර ඊට අනුරූප ව සථන ඛණ්ඩ ප්‍රදර්ශකයේ "5" ඉලක්කම දර්ශනය වේ. ඉන් පසු තත්පර 1ක පමණ කාලයක් බලා සිට නැවතත් btfsc PORTA, 0 උපදෙසට ගමන් කරයි. එ සඳහා go to Loop යන උපදෙස යොදාගෙන ඇත. එහි දී නැවතත් බොත්තම ඔබන තුරු බලා සිටී. බොත්තම එබූ පසු නැවතත් Count විචල්‍යයේ අගය 1කින් වැඩි කර එය දහය වේ නම් "0" කර නැතිනම් එලෙස ම, w රෙජිස්ටරයට මාරු කර වගුව වෙත ගොස් අදාළ අගයන් ගෙනවිත් P ORTB රෙජිස්ටරයට ලබා දේ. ඉන් පසුව නැවතත් බොත්තම එබෙන තුරු Bttsc PORTA,0 හා go to Loop යන උපදෙස් මත නොනැවතී ක්‍රියාත්මක වේ.

මෙහි දී ඔබට ගණනය කළ හැක්කේ 0 සිට 9ට පමණි. 0 සිට 99 දක්වා ගණන් කිරීම සඳහා තවත් සථන ඛණ්ඩ ප්‍රදර්ශකයක් අවශ්‍ය වේ. ඊට අදාළ පරිපථය හා ක්‍රමලේඛනය මිළඟ ලිපියෙන් බලාපොරොත්තු වන්න.

;*****Define Registers*****			
PC	equ	02h	
STATUS	equ	03h	
PORTA	equ	05h	
PORTB	equ	06h	
TRISA	equ	85h	
TRISB	equ	86h	
CounterL	equ	0Dh	
CounterH	equ	0Eh	
Count	equ	0Fh	
;*****Set up the Ports*****			
bsf	STATUS,5		;Switch to Bank 1
bsf	TRISA,0		;Set RA0 as input
clrf	TRISB		;PORT B output
bcf	STATUS,5		;Switch to Bank 0
clrf	Count		
;****Display numbers from 0****			
main	movlw	b'00111111'	
	movwf	PORTB	;Writ the number 0
;*Wait for Button press & then display*			
Loop	btfsc	PORTA,0	;Button pressed?
	goto	Loop	;If no then wait
	incf	Count,1	;If yes count up
	movlw	D'10'	;load w with ten
	xorwf	Count,0	;compare Count & 10
	btfsc	STATUS,2	;if equal result 0
	clrf	Count	;Then clear Count
	movf	Count,0	;Else continue
	call	Table	;Look Up Table
	movwf	PORTB	;Writ the number
	call	Delay	;Call the Delay loop
	goto	Loop	
;****Look Up Table for bit patterns****			
Table	addwf	PC,1	
	retlw	b'00111111'	;Number 0
	retlw	b'00000110'	;Number 1
	retlw	b'01011011'	;Number 2
	retlw	b'01001111'	;Number 3
	retlw	b'01100110'	;Number 4
	retlw	b'01101101'	;Number 5
	retlw	b'01111101'	;Number 6
	retlw	b'00000111'	;Number 7
	retlw	b'01111111'	;Number 8
	retlw	b'01100111'	;Number 9
;***** Delay Loop *****			
Delay	decfsz	CounterL,1	
	goto	Delay	
	decfsz	CounterH,1	
	goto	Delay	
	return		
	end		

රූප සටහන අංක 2

මොරටුව විශ්ව විද්‍යාලයේ විද්‍යුත් හා විදුලි සංදේශ අංශයේ ගාමිණී ජයසිංහ/කෝලින ධර්මප්‍රිය